

## USING FINANCIAL PLANNING LANGUAGES FOR SIMULATION

Paul Gray

Edwin L. Cox School of Business  
Southern Methodist University  
Dallas, Texas

### Abstract

In recent years, a large number of financial planning languages have been developed that contain capabilities for fixed time interval simulation. These languages differ from ordinary programming languages and from existing simulation languages in that they are much more English-like, produce "spread sheet" output, have built-in "what if" and other capabilities for sensitivity analysis, and are simple, easy to learn, and easy to communicate with other professionals and management.

This tutorial describes the capabilities of these languages for simulation. These capabilities include, among others, means for generating probability distributions, for generating scenarios, for providing statistical outputs on variables as a function of time, and graphical presentation of output distributions. Typical models written in one of these languages, (the Interactive Financial Planning System) are used as illustrations.

### INTRODUCTION

The use of the Interactive Financial Planning System (IFPS) for modeling was described by Tom Shriber in another tutorial presented at this Winter Simulation Conference (1). This paper discusses the simulation capabilities of financial planning languages, with particular emphasis on IFPS.

Financial planning languages are an important methodological innovation of the 1970's. They were devised in response to the need for allowing senior managers to build and use models without intermediaries. At present there are several dozen such languages available. At the simplest level, VISICALC and its derivatives provide simple spread sheets at a cost of a few hundred dollars. At the most complex level, languages such as EXPRESS provide enormous modeling and simulation capabilities at costs on the order of a quarter of a million dollars. IFPS is in the middle of this range; it is widely used and available (mid-1982) at over 135 universities and colleges. Although most of the

financial planning languages started out as deterministic languages that added stochastic features later, the Interactive Financial Planning System was developed originally by a group then at the University of Texas as a simulation tool for risk analysis.

As described by Shriber (1), IFPS is written in near-English form and has strong "WHAT IF" capabilities. In fact, it is possible to write simulations initially in deterministic form and then introduce uncertainty through the WHAT IF feature. IFPS has three properties that are important for simulation.

1. It is a column-oriented language, where the columns typically represent successive time periods. Therefore it is a language for fixed-time-interval rather than next-even simulation.
2. It is a non-procedural language, like DYNAMO. That is, you can use variables before you define them and the language performs the needed fix-up.
3. It allows the generation of scenarios as well as providing built-in probability distributions and statistical outputs designed for simulation experiments.

The sections that follow describe:

1. The built-in and user defined probability distributions
2. The scenario generation capability
3. The simulation options available
4. Interpretation of the output available
5. Three examples of increasing complexity involving business decisions
6. A discussion of the equivalence between IFPS and DYNAMO.

This paper concludes with a general discussion of the simulation capabilities provided by typical financial planning languages.

Proceedings of the 1982  
Winter Simulation Conference  
Highland \* Chao \* Madrigal, Editors

82CH1844-0/82/0000-0675 \$00.75 © 1982 IEEE

Financial Planning Languages (continued)

PROBABILITY DISTRIBUTIONS

IFPS provides four built-in distribution functions:

- UNIRAND(a,b) Uniform distribution between a and b.
- TRIRAND(a,b,c) Triangular distribution between a and c, maximum at b.
- T109ORAND(a,b,c) Triangular distribution with 10th and 90th percentiles at a and c, maximum at b.
- NORRAND(m,s) Normal distribution with mean m, standard deviation s.

Other distributions can be defined in terms of their density functions by using GENRAND or their cumulative distributions CUMRAND. GENRAND approximates the density function as a series of straight lines. A typical specification might be:

$$X = \text{GENRAND}(0,0,1,1,2,3,3,6,4,7,5,7,6,5,10,0)$$

where (0,0),(1,1),(2,3) etc. represent successive pairs of values of X (with the X's in increasing order) and the random variable given X. Whereas the number of points used for GENRAND depends on the degree of linear approximation, in CUMRAND you specify precisely 11 points; for example:

$$X = \text{CUMRAND}(0,3,5,6,7,7.5,8,8.5,9,10,12)$$

where the points represent the X values corresponding to the 0th, 10th, 20th...90th, and 100th percentile.

A particular quantity (such as rental cost) may be uncertain initially but once determined stays fixed, whereas another quantity (such as sales volume) may fluctuate from period to period. IFPS allows you to model both kinds of situations. If you use UNIRAND, TRIRAND, etc. IFPS picks a value once and keeps that value in all following periods; however, if you add an R to any of these definitions of probability distribution (e.g., UNIRANDR,TRIRANDR) you will obtain a different value from the distribution each period. You can combine the two forms. For example,

$$\text{SALES VOLUME} = \text{TRIRAND}(800,1000,1100)*\text{TRIRANDR}(0.97,1,1.03)$$

picks the first period sales volume from a triangular distribution between 800 and 1100 that represents nominal sales level; for example, 900. It then uses that value in all periods. However, it introduces perturbations around this nominal value between 97 and 103%; that is, between 873 and 927.

EXAMPLE: SIMPLE MODEL WITH UNCERTAINTY

Figure 1 shows a simple IFPS model. A new product is to be marketed and a simple forecast model is built. There is uncertainty about the level of initial demand, and the rate of growth of sales, fixed expenses, and variable expenses. These uncertainties are modeled by uniform and triangular

distributions. For example, initial sales are estimated most likely to be 100 but to be triangularly distributed between 90 and 120. The sales growth rate is nominally 25%; however, growth changes each period in a range from 20% above to 20% below the nominal.

FIGURE 1. SIMPLE MODEL WITH UNCERTAINTY

```

10 COLUMNS 1-6
20 SALES=INITIAL SALES,PREVIOUS SALES*
21 SALES GROWTH RATE
30 EXPENSES=75,PREVIOUS EXPENSES + 25
31 +FIXED EXPENSE GROWTH +VARIABLE EXPENSE GROWTH
40 NET INCOME=SALES-EXPENSES
50 *
60 *
70 *
80 INITIAL SALES=TRIRAND(90,100,120),0
90 SALES GROWTH RATE=1.25*UNIRANDR(0.8,1.2)
100 FIXED EXPENSE GROWTH =UNIRAND(-2,4)
110 VARIABLE EXPENSE GROWTH=UNIRANDR(-3,1)
    
```

SCENARIO GENERATION

Scenarios based upon the model shown in Figure 1 can be obtained by solving the model with individual variables set to:

HIGH      LOW      MEAN      MOSTP

where HIGH and LOW refer to the ends of the range (2.5 standard deviations in the case of NORRAND) and MEAN and MOSTP refer to the average and mode of the distribution. A typical example is shown in Figure 2, where a best case is presented: high initial sales and sales growth rate and low fixed and variable expense growths. By generating a variety of scenarios, it is possible to bound the range of outcomes anticipated.

FIGURE 2. BEST CASE SCENARIO FOR SIMPLE MODEL WITH UNCERTAINTY

? high initial sales,sales growth rate						
? low fixed expense growth,variable expense growth						
? all						
	1	2	3	4	5	6
SALES	120	180	270	405	607.5	911.3
EXPENSES	75	95	115	135	155	175
NET INCOME	45	85	155	270	452.5	736.3
INITIAL SALES	120	0	0	0	0	0
SALES GROWTH RATE	1.500	1.500	1.500	1.500	1.500	1.500
FIXED EXPENSE GROWTH	-2	-2	-2	-2	-2	-2
VARIABLE EXPENSE GROWTH	-3	-3	-3	-3	-3	-3

MONTE CARLO SIMULATION

Scenarios are deterministic outcomes and, in effect, are special WHAT IF cases of a deterministic model. If you ask IFPS for MONTE CARLO, the language automatically performs a simulation and provides you with statistical output. You must tell IFPS what

analysis you want performed, specifically:

1. the number of iterations (100 is the default value)
2. for which variables (and in which time periods) you want printout. If you do not specify time periods, you receive printout only on the last period.
3. the output format (e.g., histogram, frequency distribution).

Judicious selection is called for here, because it is easily possible to generate very long runs that run up big computer bills or exceed available storage. Specific output options for statistical analysis are:

HIST	Generates a histogram (and frequency data) for the variable
FREQ	Generates a cumulative observed probability distribution
NORM	Generates a table of normal probability percentile values based on the observed mean and standard deviation of the variable
COMB	Generates both FREQ and NORM for the variable.

#### INTERPRETATION OF RESULTS

MONTE CARLO results are printed in a format that differs from the usual spread sheet output of the SOLVE command. Figure 3 (following page) shows the simulation results for the simple IFPS model of Figure 1. You first specify MONTE CARLO followed by the number of iterations you want (100 in Figure 3). You are then successively asked for MONTE CARLO options. The first specification indicates that we want results for columns 1 and 4 through 6. The next specification asks for a histogram on sales, frequency distribution on expenses, and combined normal and frequency data on net income. The last option, NONE, indicates to IFPS that the list is complete.

The MONTE CARLO output is organized in the following sequence:

1. Normal approximation table
2. Frequency table
3. Sample statistics
4. Histogram

Within each of these categories, IFPS presents results in the order in which the variables are specified and for each variable in the order in which the columns are specified. Data are given in a fixed format for each category.

The normal and frequency tables are shown as rows rather than columns because of the multiplicity of data for each time period. The percentile points for the normal approximation and frequency table provide the "probability of a value greater than indicated." For example, in the frequency table for Sales in period 6, 90% of the observed values were above 207 and only 20% of the values were above 381.

The normal approximation table is based on the assumption that the variables come from a normal distribution whose mean and variance are equal to those

of the observed sample. (Except for the mean and variance, the observed data are not used in creating this table.)

The frequency table contains the 10th through 90th percentile points of the data, in increments of 10%. Thus, the frequency table summarizes the observations.

A fixed set of sample statistics is provided about each distribution given in the frequency table, including:

MEAN	average value of the observations
STD DEV	standard deviation of the observations
SKEWNESS	assymmetry about the mean (0 implies symmetry)
KURTOSIS	peakedness (4th moment) around the mean
10PC	lower 10 percent level of confidence interval around the mean
90PC	upper 90 percent level of confidence interval around the mean.

The skewness and kurtosis can be used to help form hypotheses about the underlying probability distribution of the output. For example, for a theoretical normal distribution, the skewness is 0 and the kurtosis is 3; if the observed values are close to 0 and 3, a normal hypothesis should be tested.

The histograms show the shape of the observed distribution. IFPS scales information so that it fits into a fixed format, which can make reading the graph appear tricky. The height of each column of asterisks shows the number of observations (y-axis value) in an interval. The numbers on the x-axis have to be used in conjunction with the values on the bottom of the plot, that is, the start, stop, and interval size values. The size of the interval tells you the size of each division on the x-axis and start and stop give the left and right end points, respectively. The values printed vertically are the midpoints of the particular intervals and are shown only every third column.

In summary, a rich set of output is provided, but it is relatively rigid in format.

#### EXAMPLE: A MORE COMPLEX NEW PRODUCT MODEL

Figure 4 (2nd page following) shows a more complex new product model than Figure 1. This model, which is essentially self-documenting, assumes that a novelty item is being introduced that will have a life of 8 quarters. The initial market is 8000 with an assumed growth rate in the total market of 3% per quarter. The initial market share is 15% with a linear growth of 1/2 percent per quarter. Overhead cost is 80% of production cost. For the model shown in Figure 4, the internal rate of return is 30% after 8 quarters, well above the company's 15% criterion.

In Figure 5 (2nd page following), we use WHAT IF statements to introduce uncertainty. Specifically, market growth is now triangular with parameters (0.9,1.02,1.08), market share growth is uniform between (-.002,.008) and overhead cost is normally distributed with mean 80% and standard deviation 5%. The results indicate that under these conditions

FIGURE 3. SIMULATION RESULTS FOR SIMPLE MODEL WITH UNCERTAINTY

MONTE CARLO 100  
 ENTER MONTE CARLO OPTIONS  
 ? HIST SALES,FREQ EXPENSES,COMB NET INCOME,NONE

NORMAL APPROXIMATION TABLE

PROBABILITY OF VALUE BEING GREATER THAN INDICATED

	90	80	70	60	50	40	30	20	10
NET INCOME									
6	4.2	40.9	67.4	90.0	111.1	132.3	154.9	181.4	218.1

FREQUENCY TABLE

PROBABILITY OF VALUE BEING GREATER THAN INDICATED

	90	80	70	60	50	40	30	20	10
SALES									
6	207	242	260	278	303	324	344	381	421
EXPENSES									
6	186.3	188.5	191.8	195.1	197.5	200.1	203.8	206.7	211.4
NET INCOME									
6	10.3	38.9	63.3	76.5	103.7	125.1	155.1	184.4	223.0

SAMPLE STATISTICS

	MEAN	STD DEV	SKEWNESS	KURTOSIS	10PC CONF	MEAN	90PC
SALES							
6	309.2	81.25	.4	2.8	298.8	319.6	
EXPENSES							
6	198.1	9.184	.1	2.0	196.9	199.2	
NET INCOME							
6	111.1	83.43	.4	2.8	100.5	121.8	

HISTOGRAM FOR COLUMN 6 OF SALES

21- 22	*
19- 20	*
17- 18	**
15- 16	***
13- 14	***
11- 12	*****
9- 10	*****
7- 8	*****
5- 6	*****
3- 4	*****
1- 2	*****

1	2	3	5
5	7	9	2
1	4	7	0

START 130.0 STOP 540.0 SIZE OF INTERVAL 41.00

although the observed mean of 200 replications is 15.17%, the investment meets the 15% internal rate of return requirement only 50% of the time. From a methodological point of view, the language allows writing a simple deterministic model and then using WHAT IF to introduce a combination of uncertainties that require simulation.

EXAMPLE: A RISK ANALYSIS MODEL

This model is based on a classic paper by David B. Hertz (2). A medium sized chemical firm is considering a \$10 million improvement program for its main plant. The improvement is expected to have a life-time of 10 years. Management has determined the key input factors in analyzing the profitability of this proposed capital investment are market size, selling price, market share, total investment, salvage value of the investment, operating costs, and fixed costs. Unfortunately, as shown in Figure 6, all of these quantities have large uncertainties associated with them. In building its forecast, the company knows that it has the option of closing the plant down if the selling price goes below the operating cost; however, it would still have to pay the fixed cost. This option is implemented in Figure 6 by use of a MAXIMUM statement. If all variables are set to their expected values, then this investment would yield a 9.9% internal rate of return.

FIGURE 4. NEW PRODUCT MODEL

```

10 COLUMN 1-8
20 *
30 *
40 *NEW PRODUCT MODEL
50 *
60 SALES VOLUME=MARKET*MARKET SHARE
70 MARKET=8000,PREVIOUS MARKET*1.03
80 MARKET SHARE=0.15,PREVIOUS MARKET SHARE +0.005
90 GROSS SALES =SALES VOLUME*UNIT PRICE
100 UNIT PRICE =2.5 FOR 4,2.75 FOR 4
110 OPERATING COST =SALES VOLUME * (PRODUCTION COST
111 + OVERHEAD)
120 PRODUCTION COST =1.25 FOR 4,1.35
130 OVERHEAD = PRODUCTION COST *0.80
135 NET INCOME =GROSS SALES -OPERATING COST
140 INVESTMENT =2500,0
150 RATE OF RETURN = IRR(NET INCOME,INVESTMENT)
160 NET PRESENT VALUE=NPVC(NET INCOME,
161 DISCOUNT RATE,INVESTMENT)
170 DISCOUNT RATE =.15
180 PERIODS 4

```

FIGURE 6. RISK ANALYSIS MODEL

```

30 *
40 * RISK MODEL FOR CHEMICAL COMPANY
50 *
60 COLUMNS 1-10
70 MARKET = TRIRAND(100000,250000,340000),PREVIOUS'
71 * TRIRAND(1,1.03,1.06)
80 SELLING PRICE = TRIRAND(385,510,575)
90 MARKET SHARE = UNIRAND(.12,.17)
100 SALES VOLUME = MARKET * MARKET SHARE
110 INVESTMENT =TRIRAND(7,9.5,10.5)*1000000,0
120 LIFE = 10
130 OPERATING COST = TRIRAND(375,435,545)
140 FIXED COST = TRIRAND(250,300,375)*1000
150 REVENUE = MAXIMUM(0,(SELLING PRICE'
151 - OPERATING COST)*SALES VOLUME)
160 NET INCOME = REVENUE - FIXED COST
170 SALVAGE VALUE =0 FOR 9,TRIRAND(3.5,4.5,5)'
171 *1000000
180 *
190 RATE OF RETURN =IRR(NET INCOME + SALVAGE VALUE,
191 INVESTMENT)
200 *

```

FIGURE 5. USE OF WHAT IF STATEMENT FOR SIMULATION

```

?WHAT IF
WHAT IF CASE 1
ENTER STATEMENTS
?MARKET = 8000,PREVIOUS*TRIRAND(0.90,1.02,1.08)
?MARKET SHARE = .15,PREVIOUS+UNIRAND(-.002,.008)
?MONTE CARLO 200
ENTER MONTE CARLO OPTIONS
?FREQ RATE OF RETURN,NET PRESENT VALUE, NONE
****WHAT IF CASE 1 ****

```

FREQUENCY TABLE

PROBABILITY OF VALUE BEING GREATER THAN INDICATED

90	80	70	60	50	40	30	20	10
----	----	----	----	----	----	----	----	----

RATE OF RETURN

8	.070	.088	.106	.128	.147	.160	.181	.204	.225
---	------	------	------	------	------	------	------	------	------

NET PRESENT VALUE

8	-199	-157	-111	-57	-8.2	27	81	139	192
---	------	------	------	-----	------	----	----	-----	-----

SAMPLE STATISTICS

MEAN STD DEV SKEWNESS KURTOSIS 10PC CONFMEAN 90PC

RATE OF RETURN

8	.1457	.0613	.1	2.5	.1402	.1513
---	-------	-------	----	-----	-------	-------

NET PRESENT VALUE

8	-9.02	155	.1	2.5	-23.1	5.04
---	-------	-----	----	-----	-------	------

Figure 7 (following page) shows the much richer set of data obtained from the output of a simulation with 200 replications. We see that even by year 10, the net income can fluctuate widely between a loss of over \$300,000 and a profit of over \$4 million. Furthermore, the rate of return on this investment ranges between -12% and +42.2% with only a 40% chance that it will exceed 15%. The histogram indicates that there is a significant risk of losing money. If you are a manager faced with this situation you may well not want to take the risk even though the project's expected payout is above 10%.

Financial Planning Languages (continued)




FIGURE 7. SIMULATION RESULTS FOR RISK ANALYSIS

```

?MONTE CARLO 200
ENTER MONTE CARLO OPTIONS
?HIST NET INCOME, HIST RATE OF RETURN, NONE
      FREQUENCY TABLE
PROBABILITY OF VALUE BEING GREATER THAN INDICATED
      90  80  70  60  50  40  30  20  10
NET INCOME
10 -302 -184 489 1324 1866 2378 3293 4252 5175*1000
RATE OF RETURN
10 -.120 -.090 -.005 .088 .157 .201 .280 .380 .455

```


SAMPLE STATISTICS

```

      MEAN STD DEV SKEWNESS KURTOSIS 10PC CONFMEAN 90PC
NET INCOME
10 2129150 2057517 .5      2.1      1942925 2315375
RATE OF RETURN
10 .1570 .2186 .3      2.1      .1371 .1768

```

HISTOGRAM FOR COLUMN 10 OF NET INCOME



	1	3	4	6
-	5	2	8	5
9	5	0	5	0
5	5	7	8	9
7	6	0	4	8
6	3	3	3	3
7	3	3	3	3

```

START -371000.0      STOP 7886000
SIZE OF INTERVAL 550466.7

```

THE EQUIVALENCES BETWEEN IFPS AND DYNAMO

DYNAMO (3) has been with us for nearly 25 years. It is now a mature language that has had wide application in simulation. Although thought of as a continuous language, it is really a fixed-time-interval language since it slices time into fixed intervals, DT. It is also a non-procedural language. As a result, anything that can be written in DYNAMO can be translated very readily into IFPS (4). Figure 8 shows the simple retail sector model given in Pugh's Dynamo Users Manual (3) and Figure 9 (following page) shows the equivalent IFPS model. As shown in (4), the two versions give identical output. Comparing the two sets of code shows that the IFPS model is easier to read and understand. The UOR.K = UOR.J + DT\*(RRR.JK - SSR.JK) statements are gone. Time relations are explicit in terms of PREVIOUS and FUTURE and initial values are shown on the same line as the basic relations. The same ideas are expressed in a form that is understandable by client and manager.

FIGURE 8. DYNAMO MODEL FOR SIMPLE RETAIL SECTOR (after Pugh (3))

```

* SIMPLE RETAIL SECTOR
L UOR.K=UOR.J+DT*(RRR.JK-SSR.JK)
N UOR=DFR*RRR
R SSR.KL=UOR.K/DFR.K
A DFR.K=TABLE(TDFR,IAR.K/RSR.K,4,12,4)
T TDFR=1.5,1,.75 WK
A RSR.K=SMOOTH(RR.JK,DRR)
C DRR=8 WKS
A IDR.K=AIR*RSR.K
C AIR=8 WKS
R PSR.KL=RSR.K+(IDR.K-AIR.K)/DIR
R SRR.KL=DELAY3(PSR.JK,DTR)
C DTR=6 WKS
L IAR.K=IAR.J+DT*(SRR.JK-SSR.JK)
N IAR=IDR
NOTE INPUT
R RRR.KL=RRI+STEP(STH,5)
C RRI=1000 ITEMS/WK
C STH=100 ITEMS/WK

```

FIGURE 9. IFPS RETAIL SECTOR MODEL

```

10 COLUMNS 0-30
20 UNFILLED ORDERS=INITIAL ORDERS,PREVIOUS UNFILLED'
21 ORDERS +(TIME INTERVAL)*(PREVIOUS NEW ORDERS'
22 -PREVIOUS SHIPMENT OUT)
30 SHIPMENT OUT=UNFILLED ORDERS/ORDER FILL DELAY
40 ORDER FILL DELAY +INTERPOLATION ON'
41 (RATIO,4,1.5,8,1,12,0.75)
50 RATIO=ACTUAL INVENTORY/SMOOTHED ORDERS
60 SMOOTHED ORDERS=NEW ORDERS, PREVIOUS SMOOTHED'
61 ORDERS+(TIME INTERVAL)*DELAY SMOOTHING CONSTANT'
62 *(PREVIOUS NEW ORDERS-PREVIOUS SMOOTHED ORDERS)
70 DESIRED INVENTORY=NO OF WEEKS SUPPLY *'
71 SMOOTHED ORDERS
80 PURCHASES=SMOOTHED ORDERS+(DESIRED INVENTORY'
81 -ACTUAL INVENTORY)/INVENTORY DELAY CONSTANT
90 SHIPMENTS IN=DELAY3
100 ACTUAL INVENTORY=DESIRED INVENTORY, PREVIOUS'
101 ACTUAL INVENTORY +(TIME INTERVAL)*'
102 PREVIOUS SHIPMENTS IN -PREVIOUS SHIPMENTS OUT)
110 *
120 *ASSUMPTIONS
130 *
140 NEW ORDERS = 1000 FOR 20,1100
150 INITIAL ORDERS=ORDER FILL DELAY*NEW ORDERS,0
160 DELAY SMOOTHING CONSTANT=1/8
170 NO OF WEEKS SUPPLY=8
180 INVENTORY DELAY CONSTANT=4
190 TRANSPORTATIONDELAY=6
200 *
210 *DELAY3 COMPUTATION
220 *
230 DELAY3=LEVEL3/DELAY
240 LEVEL3=DELAY*PURCHASES,PREVIOUS LEVEL 3 +'
241 (TIME INTERVAL)*(PREVIOUS LEVEL2/DELAY'
242 -PREVIOUS DELAY3)
250 LEVEL 2=LEVEL3,PREVIOUS LEVEL2 +(TIME INTERVAL)'
251 *(PREVIOUS LEVEL1/DELAY-PREVIOUS LEVEL2/DELAY)
260 LEVEL 1=LEVEL3,PREVIOUS LEVEL1 +(TIME INTERVAL)'
261 *(PREVIOUS PURCHASES - PREVIOUS LEVEL1/DELAY)
270 DELAY=TRANSPORTATIONDELAY/3
280 *
290 *TIME INTERVAL IN WEEKS
300 *
310 TIME INTERVAL = .25

```

COMPARISON OF SIMULATION CAPABILITIES  
OF FINANCIAL PLANNING LANGUAGES

The number of commercial financial planning languages available today is over 40 if you limit yourself to mainframe and minicomputer systems; many more if you include VISICALC and similar spreadsheet languages available for microcomputers. Although the microcomputer languages at present do not have stochastic simulation capabilities, most (but not all) of the mainframe/minicomputer languages do have such capabilities. A comparison of eight of these languages (CUFFS, EMPIRE, EXPRESS, FCS, IFPS, REVEAL, SIMPLAN, XSIM) by Naylor and Mann (5) indicates the range of capabilities they provide other than simulation. Capabilities include planning and budgeting, statement consolidation, database management, report generation, graphics, mathematical modeling and optimization, forecasting, financial and statistical subroutines. Each of the languages has almost all these capabilities to greater or lesser extent. The main differences between languages are

in what they emphasize and in whether they are procedural or non-procedural. The merits of procedural vs. non-procedural languages is a subject of heavy debate among the developers of these languages, with merit on both sides. Simulation capabilities offered vary from language to language; for example, the number of built-in probability distributions. (CUFFS is the only one of these languages that does not provide stochastic simulation capability; however, it can be used for deterministic simulation of the DYNAMO type.)

In my own review of these eight languages and a ninth one, MODEL, it is clear that the developers of these languages view simulation principally as a risk analysis tool. They have not exploited the simulation capabilities of their languages; usually providing the level of output described above for IFPS.

CONCLUSIONS

The natural question for the professional in simulation to ask, is: "Why use financial planning languages for simulation?", or more bluntly "What's in it for me?". The answer is threefold:

- (1) they are excellent tools for fixed-time interval simulation.
- (2) they are integrated languages that remove many of the drudgery tasks in programming.
- (3) they are written in a form that makes it easy for the simulation professional to communicate the results obtained to the client and for the university professor to communicate underlying concepts of simulation and modeling to the student.

In their present state of development, financial planning languages do indeed have limitations for simulation. However, they are forerunners of the English-like, easy to read and write computer languages toward which we are all heading. For many applications, particularly in business, they are really the language of choice.

BIBLIOGRAPHY

1. Shriber, T. S. "IFPS Tutorial," Proceedings of the 1982 Winter Simulation Conference, December 1982.
2. Hertz, D. B. "Risk Analysis in Capital Investment," Harvard Business Review, January-February 1964.
3. Pugh A. L., III. Dynamo Users Manual, Fifth Edition, The MIT Press, Cambridge, MA, 1976.
4. Gray, P. "On the Equivalence Between IFPS and DYNAMO," Southern Methodist University Working Paper, September 1982.
5. Naylor, T. H. and M. H. Mann. "A Comparison of Eight Planning and Modeling Software Systems," Corporate Economics Program, Duke University, June 25, 1981.